



以上方法中,第 1 的气相接枝法由于单体浓度较低,因而与基质表面活性点接触的机率减小,所以接枝率较低。而第 2 液状法的聚合度可以提高,但同时均聚物也增多,必须设法除去。第 3 方法由于空气中氧的阻聚作用,直接枝效率也较低些。

以等离子体接枝共聚方法改性高分子材料,可以获得较为耐久的变性效果;如以乙烯基吡啶气相接枝聚丙烯或聚乙烯薄膜,可以改善表面润湿性。

四、等离子体引发聚合

与上述等离子体聚合不同,“单体”不是进入等离子体系统内,而是经过等离子体预照射后“单体”内生成活性种,以此活性种再进行均化聚合。如甲基丙烯酸甲酯,经等离

子体引发后聚合,可以获得分子量达到数百万的超高分子量聚合物。等离子体还可引发聚合固态的三氧杂环己烷,获得高结晶度、高弹性模量的材料。以往三氧杂环己烷是以 γ 射线或 X 射线预辐照法进行引发聚合的,其辐照剂量高,且得率低。环状硅氧烷也有类似的结果。

结 论

以低温等离子体技术改性高分子材料,可以减少三废,且不影响材料的本体性质,其设备较为简便,是颇为有前途的一种加工工艺。但低温等离子体的各种参数的诊断和控制仍需要进一步完善化,以提高处理加工的重演性。总之,低温等离子体技术在高分子材料改性方面的应用尚处于初始开发阶段,还有很多工作等待人们去做。

微机软件在化工计算中的应用(一)

朱 曾 用(上海市化工装备研究所)

为了帮助广大读者应用个人电脑进行计算机辅助设计、数据处理等方面工作,我们将陆续刊登以 BASIC 语言编写的源程序,这些程序都已在 APPLE-II 型微机上通过计算。当然,读者如用其他型号的微机,只需略作修改也可应用。每期刊登一至若干个源程序,并介绍其计算原理与使用方法,以便读者不仅能应用此程序计算自己的课题,并且能逐步掌握一些技巧来自己编制新的程序。对每个程序都附有一个计算例题,这不仅可以帮助读者进一步了解其原理及使用方法,而且可以根据答案来检查你输入程序的正确性。

本讲座以分离工程、传热工程、流体输送

等方面应用为主。限于篇幅,刊登的程序不能太大。加以 BASIC 语言是一种边编译边执行的语言,计算时间较长,不适于较大型的计算,读者如需要更为复杂的化工方面应用软件,可以来信联系,我们还在移植与编制应用 FORTRAN 语言在 IBM/PC 微机上通过计算的应用软件。

一、多组分蒸馏塔简捷计算

多组分蒸馏塔简捷计算程序可用在精确计算前初步确定设计参数,进行技术经济比较,为精确计算时节省大量时间。个别情

况下,也可直接用来求取设计参数,其中主要应用 Underwood 求最小回流比, Fenske 求最小理论塔板数,以及 Gilliland 求理论塔板数的公式。

Underwood 求最小回流比 R_{\min} 公式包括下列两式:

$$R_{\min} = \left(\sum_1^K \frac{\alpha_i D_i}{\alpha_i - \theta} \right) - 1 \quad (1)$$

$$1 - q = \sum_1^K \frac{\alpha_i F_i}{\alpha_i - \theta} \quad (2)$$

其中 α 为相对挥发度,可以取塔顶、塔底及进料板处的几何平均值:

$$\alpha = (\alpha_{\text{顶}} \cdot \alpha_{\text{底}} \cdot \alpha_{\text{进料}})^{1/3} \quad (3)$$

q 可看成进料处液相分子分率,对于以露点进料则等于零,以泡点进料则为 1。

B 为塔底馏出物中各组分的分子分率, D 为塔顶馏出物中各组分的分子分率, F 为进料中各组分的分子分率, θ 为迭代因子,下标 i 为组分顺序号, k 为组分数(本程序中最多可至 8 个组分)

Fenske 求最小理论塔板数 N_{\min} 为:

$$N_{\min} = \lg \left(\frac{D_{1k} B_{hk}}{D_{hk} B_{1k}} \right) / \lg(\alpha_{1k}) \quad (4)$$

其中下标 hk 代表重关键组分, lk 代表轻关键组分。

Gilliland 求理论塔板数 N 的关系,按 Eduljee 关联的公式为:

$$\frac{N - N_{\min}}{N + 1} = 0.75 - 0.75 \left(\frac{R - R_{\min}}{R + 1} \right)^{0.5668} \quad (5)$$

(R 为回流比)

应用 Fenske 公式还可求最小精馏段理论塔板数:

$$N_{D\min} = \lg \left(\frac{D_{1k} \cdot F_{hk}}{D_{hk} \cdot F_{1k}} \right) / \lg(\alpha_{1k}) \quad (6)$$

由此可得精馏段理论塔板数:

$$N_D = N - \frac{N_{D\min}}{N_{\min}} \quad (7)$$

故进料板序号为 $N - N_D$

使用此程序时应已知进料、塔顶、塔底产品的各组分分子分率。并确定轻、重关键组

分,轻关键组分与重关键组分相邻,程序中各组分次序由轻向重依次排列。并以重关键组分的相对挥发度为 1,计算出其他组分的相对挥发度和平均相对挥发度。

程序中由第 20 句读入组分数 k ,进料液相分子分率 q 及重关键组分与轻关键组分的顺序号 hk, lk 。第 30~50 句依次读入各组分之进料、塔顶、塔底产品之分子分率和平均相对挥发度。第 60~150 句应用对分法迭代求解公式(1)和(2)。第 160~210 句为按式(1)求最小回流比,并打印出此值,然后使用时可根据此最小回流比值,确定实际使用回流比,并用键盘输入。程序便进一步在第 220 句和 240~260 句按式(5)、(6)、(7)计算最小理论塔板数、理论塔板数和精馏段理论塔板数。在第 280~320 句中打印出结果,包括最小回流比、最小塔板数、理论塔板数和进料板序号(自下向上计)。输入数据可在第 400 句以后顺序存放。

例题 试设计一个分离丁烷的蒸馏塔,其相应的输入参数假设与计算如下:

组分名称	i	F_i	D_i	B_i	α_i
C_3	1	0.05	0.102	0.0	4.99
$i-C_4$	2	0.15	0.301	0.004	2.62
$n-C_4$	3	0.25	0.473	0.033	2.02
$i-C_5$	4	0.20	0.069	0.327	1.0
$n-C_6$	5	0.35	0.055	0.636	0.86

重关键组分为 $i-C_5$,轻关键组分为 $n-C_4$,进料状态为泡点进料,故 $q=1$ 。采用的回流比为最小回流比的 1.25 倍。

程序及计算结果如下:

```

JLIST
10 DIM F(8),D(8),B(8),AF(8)
20 READ K,Q,HK,LK
30 FOR I=1 TO K
40 READ F(I),D(I),B(I),AF(I)
50 NEXT I
60 X1=1:X2=AF(LK):SI=(X1+X2)/2
70 F1=0:F2=0
80 FOR I=1 TO K

```

```

90 FX=AF(I)*F(I)/(AF(I)-SI)
95 F1=F1+FX
110 NEXT I
120 FS=F1+Q-1
130 IF ABS(FS)<0.0001 THEN 160
135 IF FS>0 GOTO 145
140 X1=SI:SI=(SI+X2)/2: GOTO 70
145 X2=SI:SI=(SI+X1)/2
150 GOTO 70
160 RM=0
170 FOR I=1 TO K
180 RM=AF(I)*D(I)/(AF(I)-SI)+RM
190 NEXT I
200 RM=RM-1
210 PRINT "RM=";RM
220 NM=LOG(D(LK)*B(HK)/D(HK)/B(LK))/
    LOG(AF(LK))
230 PRINT "INPUT R":INPUT R
240 FR=0.75-0.75*((R-RM)/(R+1))^0.5668
250 N=(NM+FR)/(1-FR)
260 NN=LOG(D(LK)*F(HK)/D(HK)/F(LK))/
    LOG(AF(LK))
270 ND=N*NN/NM
280 PRINT "THE RESAULT OF
    CALCULATIONS:"
290 PRINT"MINIMUM REFLUX RATIO=";RM
300 PRINT "MINIMUM THEORETICAL
    PLATES=";NM
310 PRINT"THEORETICAL PLATES=";N
320 PRINT"FEED PLATE NO.=";N-ND
395 DATA 5,1,4,3
400 DATA .05,.102,0,4.99
410 DATA .15,.301,.004,2.62
420 DATA .25,.473,.033,2.02
430 DATA .2,.069,.327,1
440 DATA .35,.055,.636,.86
JRUN
RM=.93727376
INPUT R
?1.16896
THE RESAULT OF CALCULATIONS:
MINIMUM REFLUX RATIO=.93727376
MINIMUM THEORETICAL PLATES=5.99979579
THEORETICAL PLATES=14.180468
FEED PLATE NO.=8.45964354

```

二、二元系蒸馏塔逐板计算

此程序适用于二元系逐板计算各板汽液相组成,在已确定进料量、理论塔板数及汽化量情况下,根据要求塔顶产品组成,求出塔顶产品流量,故也确定了回流量。这是一个操作型计算程序,但由于在计算机上快速计算,可以多次输入不同参数,重复计算,故也可用于设计。

程序中开始计算时,先假定一个馏出物流量与进料量之比值 D ,由物料衡算,在精馏段可写出:

$$L/V = 1 - D/V \quad (1)$$

其中 L 为塔内液相流量与进料流量之比, V 为蒸气流量与进料量之比。

对于二元系,由已知塔顶气相轻馏分子分率 y_1 ,可以按下式逐板计算各板的气相与液相中轻馏分组成:

$$x_i = y_i / [y_i + \alpha(1 - y_i)] \quad (2)$$

$$y_{i+1} = y_i - L/V (x_{i-1} - x_i) \quad (3)$$

其中 x_i 为第 i 块板上液相轻组分分子分率, y_i 为第 i 块板上气相轻组分分子分率, α 为相对挥发度。

当 $i=1$ 时, (3) 式中用到 x_0 , 程序中假定塔顶蒸气全冷凝后回流。故 $x_0 = y_1$ (第 30 句), 在进料板处由物料衡算可得:

$$y_{NF+1} = y_{NF} + [1 + (1 - D)/V] \cdot x_{NF} - kx_{NF-1} - z/v \quad (4)$$

其中 z 为进料中轻组分分子分率, N_F 为进料板以上的塔板数。 $K = 1 - D/V$ 。

在提馏段有:

$$L/V = 1 + (1 - D)/V \quad (5)$$

故由 y_{NF+1} 开始,按式(2)及(3)又可逐板计算气液相中轻馏分组成,直至最后一块板即塔釜或再沸器。

由全塔物料衡算可求出塔顶馏出物流量与进料量之比值 D_c :

$$D_c = (z - x_{NB}) / (y_1 - x_{NB}) \quad (6)$$

其中 N_B 为总塔板数。

比较 D_c 是否与原假定 D 相等, 如不等, 则按下式重新设定 D :

$$D = D + K_D(D_c - D) \quad (7)$$

K_D 为迭代反馈增益比值, 也即使新假定值之增量与误差成正比。如计算时收敛太慢, 可增大比值 K_D , 太快而形成振荡发散, 则可减小 K_D 值, 此迭代过程见第40~180句。

程序中由于(2)、(3)式多次重复应用, 故编成子程序(第250~270句)调用。在运算时不断打印出 D_c 值, 以便判断是收敛还是发散。程序在 $(D_c - D)/D_c$ 绝对值小于某一小值后结束。此值可根据需要精度和计算速度来选择, 这里取 0.0001(见第170句)。

例题: 已知汽化量与进料量之比 $V=3$, 为要求塔顶气相中轻馏分分子分率为 0.928, 进料中轻馏分分子分率为 0.5, 平均相对挥发度为 1.5, 进料板以上塔板数为 10, 总塔板数为 20, 求馏出物流量与进料量之比值。

程序及计算结果如下:

```

]LIST
10 DIM Y(100), X(100)
20 READ Y(1), D, V, NF, NB, Z, A, KD
25 DC=D
30 X(0)=Y(1)
40 D=D+KD*(DC-D)
50 K=1-D/V
60 FOR I=1 TO NF
70 GOSUB 250
80 NEXT I
90 Y(NF+1)=Y(NF)+(1+(1-D)/V)*X(NF)-K
    *X(NF-1)-Z/V
100 K=1+(1-D)/V
110 FOR I=NF+1 TO NB-1
120 GOSUB 250
130 NEXT I
140 X(NB)=Y(NB)/(Y(NB)+A*(1-Y(NB)))
150 DC=(Z-X(NB))/(Y(1)-X(NB))
160 PRINT DC
    
```

```

170 IF ABS((DC-D)/DC)<0.0001 THEN 190
180 GOTO 40
190 PRINT "THE RESULT OF
    CALCULATIONS"
200 PRINT "D="; D
210 PRINT "NO. "; " X"; " Y"
220 FOR I=1 TO NB
225 PRINT " "; I; " "X(I); " "Y(I)
230 NEXT I
240 END
250 X(I)=Y(I)/(Y(I)+A*(1-Y(I)))
260 Y(I+1)=Y(I)-K*(X(I-1)-X(I))
270 RETURN
300 DATA .928, 5, 3, 10, 20, .5, 1.5, .5
]RUN
    
```

```

.473087621
.490433047
.488103129
.488336405
    
```

THE RESULT OF CALCULATIONS
D = .488295779

NO.	X	Y
1	.895752896	.928
2	.858506234	.901001604
3	.816659949	.869817405
4	.771083867	.834782241
5	.723094463	.796624362
6	.674328279	.756445966
7	.626530045	.715617223
8	.581310063	.675598881
9	.539939641	.637739141
10	.503235835	.603102886
11	.472647671	.573451296
12	.436692463	.537645767
13	.395743004	.495557749
14	.350750479	.44762362
15	.303224767	.394956807
16	.25506662	.339324725
17	.208279258	.282952336
18	.164646275	.225184544
19	.125480705	.177109167
20	.0915128929	.131263202

